International Council for the
Exploration of the Sea

C.M. 1987/C:14
Hydrography Committee

AN ALGORITHM FOR DETERMINING THE APPROPRIATE METER WHEEL VALUES FOR SAMPLING OF STANDARD DEPTHS AT HYDROGRAPHIC STATIONS

by

J. B. Sochasky
Biological Sciences Branch
Marine Fish Division
Department of Fisheries and Oceans
Biological Station
St. Andrews, New Brunswick EOG 2X0
Canada

Abstract

A calculator (or computer) algorithm has been developed for use at sea to provide automatic calculation and tabulation of meter wheel values and standard depths for "full" hydrographic stations; it was created to streamline procedures and eliminate errors during the refamiliarization period typically required in the hydro lab at the beginning of research surveys. The algorithm is presented in three forms: a working program for the HP41C and similar calculators; a general flowchart; and as an explicit set of instructions for effectively determining meter wheel values "by hand." To benefit from the working program, the user must be able (1) to push one button to start the program running, and (2) key in the sounding depth; a three-column table with methodological reminders is then printed. Both the machine-based algorithm and the worksheet suggested should prove helpful in improving the quality and quantity of data collected at hydrographic stations.

Résumé

Un algorithme pour calculatrice (ou ordinateur) a été développé pour usage sur la mer pour fournir automatiquement le calcul et al classification des données du compteur à poulie [meter wheel] et des profondeurs standards pour les stations hydrographiques complètes. Celui-ci, fut créé pour faciliter le procédé et éliminer les erreurs durant la période de refamiliarisation au laboratoire hydrologique normalement nécessaire au débuts de croisières de recherches. L'algorithme est présenté en trois formules: un programme pour le HP41C et autres calculatrices semblables, un schéma général, et des directions détaillées pour déterminer les données du compteur à poulie [meter wheel] sans l'usage de calculatrice. Pour profiter du programme, l'usager doit pouvoir (1) pousser un bouton pour mettre le programme en marche, et (2) introduire la vraie profondeur de sondage. Un tableau de trois colonnes ayant des mémentos sur la methodologie est alors imprimé. L'algorithme pour calculatrice et la formule de travail proposée devraient tous deux être utiles pour améliorer la qualité et al quantité des données recueillies aux stations hydrographiques.

Introduction

The collection of "STD" data (Salinity and Temperature at known Depths) throughout the water column is the single most important aspect of hydrographic surveys and it is usually the main ancillary hydrographic data collected during biological surveys at sea. The invention of the reversing bottle resolved the major technological frustrations of obtaining samples at depths (see Defant 1961, Subsidiary Committee on Oceanography 1932) and the use of several such bottles attached at intervals along the hydrographic wire at a single multi-sampled station has resulted in the accumulation of enormous amounts of data on a world wide scale that has become assimilated on widely accessible data bases. The more recent technical development of "CTD" equipment that can give continuously recorded data has effected a further revolution, but its expense and the need for expert electronic servicing will likely preclude its entirely replacing the now traditional water bottle string or even becoming the dominant method within the next few decades. So, possible improvements in reversing bottle methodology remain topical.

Although the occupation of a full hydrographic station is in principle extremely simple, in fact to obtain samples from the depths actually desired, care must be taken to ensure that the bottles are placed on the hydrographic wire at the "appropriate intervals" and that they are then lowered to the appropriate position before the samples are actually taken. (The standard international depths (U.S. Hydrographic Office 1955) are presented in Table 1.) Otherwise, errors can occur that can result in loss of time, loss of precision or data that is incorrect or misleading, as misrepresenting the depth at which the actual sampling took place. Even when the depths sampled can be calculated afterwards from the readings of "unprotected" thermometers, if these have been used, it nonetheless remains desireable to sample as close to the target depths as possible.

The particular problems of full hydrocasts arise, in part at least, from the mental gymnastics required to maintain one set of values for the meter wheel readings, which <u>increase</u> from say zero for the deepest (the "bottom" or "first") bottle to some larger value for subsequent bottles, while the set of values for the target depths start at a comparatively large value for that first bottle (corresponding normally to a maximum depth close to the sea bed) and then <u>decrease</u> for subsequent bottles (see Fig. 1).

Inconsistencies or errors can arise from any of the several steps required:

- ensure the sounding is the true depth from the surface and not just the depth below the hull

- position the weight consistently just below the surface

- make the appropriate adjustment for the distance between the hydrolab and the surface (i.e. between where the bottles are attached at the hydrolab and the hydro weight just below the surface, typically about 5 m)

- make a consistent adjustment for roll compensation (typically 1 m) so the weight doesn't strike bottom

Fig. 1. A "full hydro cast" for a sounding of 120 m showing intervals, target depths, meter wheel values, and "protocol" variables.



(from Koeller, 1981)

- zero the meter wheel at the appropriate point

- calculate the appropriate intervals between bottles

- attach the bottles and messengers at the appropriate intervals

- lower the entire cast to the appropriate final depth

- correct, if necessary, for wire angle

- after the appropriate delay (3-6 min), release the final messenger

- retrieve the bottles without confusing which bottle was associated with which depth

Errors in these steps are likely to occur most frequently at the beginning of a cruise when inexperienced, or out-of-practice scientific crew members are involved; but errors may also occur at any time when marginal working conditions, fatigue and even overfamiliarity with a procedure that is, even so, practised only infrequently each year can result in lapses of concentration. Such errors can be costly; the data may be useless resulting in incomplete results, or time consuming data editing may be required. Any method, or protocol substantially minimizing such loss could easily be justified in relation to the very high cost of sea time.

I've surveyed several written guidelines on the subject, attended training seminars, and discussed the problem with experienced workers; there seems to be no straightforward "on-board" guide that would help assure the samples collected actually are from the desired depths, and that they were collected in a consistent way from cruise to cruise, watch to watch, and technician to technician.

The available guides usually ignore the specifics of the problem: "the sampling bottles and thermometers are lowered ... to predetermined depths ..." (Defant 1961); "the bottles ... [are] lowered to the proper depths ..." (Subsidiary Committee on Oceanography 1932); "... the bottles ... [are] attached at appropriate intervals ..." (Sverdrup et al. 1942); "... load the wire with bottles at predetermined intervals ..." (Weyl 1970); "... several bottles are attached at intervals along the wire ..." (U.S. Hydrographic Office 1955). This lack of an explicit, "conventional" or even "stereotyped" step-by-step approach is the normal situation.

The best written guide available to us (Koeller 1981) is an internal document that provides an outline of the calculations and procedures which concludes by saying "The above instructions only summarize some procedures which if not done carefully, will lead to errors and inaccuracies." Whereas, other sources recommend zeroing the meter wheel when the first (deepest) bottle is attached with the hydro weight in the water (U.S. Hydrographic Office 1955; Sverdrup et al. 1942; Weyl 1970) or when the first bottle has been lowered to the water's surface (Subsidiary Committee on Oceanography 1932), Koeller (1981) presents the two points of zeroing as alternates, discussing each essentially without preference.

Experience indicates that none of the guides available can be considered fool-proof; each cruise begins with a more or less awkward period of refamiliarization with procedures and that mistakes and inconsistencies may still arise.

This communication is one in a series of efforts within the Marine Fish Division, Dept. of Fisheries and Oceans (e.g. Koeller 1981; Perry 1987, Gavaris et al., in prep.) to streamline procedures at hydrographic stations so as to minimize errors and maximize the quality of the data collected. It presents an algorithm and associated methodology which, if properly followed, virtually eliminate the possibility of inappropriately positioning the reversing bottles during a full hydro cast.

The algorithm reported here was first thought of at sea during just such a period of refamiliarization. For convenience, it was developed for a widely used programmable calculator ("HP41C") which was already onboard to run a program to calculate the correction needed when the hydro wire angle deviates from the vertical. This relationship to the calculator is reflected in the particular structure of the program developed. In other situations, the structure might have reflected mechanisms based on spreadsheet or matrix manipulation software or on the logic of an available computer language. The general user-related features would, however, likely be similar to those of the program described here regardless of the underlying structure.

The program as seen by the user

The complexities of the working program (Appendices 1, 2) are transparent to the user. On the HP41C, the program can be assigned to a single key for execution; the user starts running the program by pushing that key. The program then prompts for the sounding depth, and once this has been entered, the user can begin to attach his first bottle.

The program, which must have been initialized at the beginning of the cruise (another single key to push and 3 prompts for "protocol" variables), generates output on a compatible printer which uses 5.7 cm (2.25 in) paper tape (Fig. 2). After it prompts for the sounding depth; it prints reminders to "set weight to surface," "set meter wheel to zero;" and "attach first bottle;" then, a 3-column table showing the bottle attachment order, the corresponding meter wheel readings and the corresponding target depths, all appropriate for the particular sounding; and finally, another reminder to lower the entire cast to the final depth (Fig. 3).

The advantages are several. The difficulties of determining the appropriate intervals and attachment points for each bottle are eliminated. Hence its use largely eliminates the "refamiliarization" period and the related variability from technician to technician and cruise to cruise. Bottle attachment can begin while the table is being printed, reducing the time spent on each station.

The paper tape can be a permanent data record for the station. All in all, the use of the program improves both the quality and quantity of the hydrographic data collected.

The complete working "program" (it's in fact a set of programs or subroutines) for the HP41C and a full description of its features is presented in Appendix 1. The general form of the algorithm is presented in

Fig. 2. The HP41C and printer in use.



Fig. 3. Printer output for a sounding depth of 120 m. Compare Fig. 1, Fig. 4 and the "worksheet" (Fig. 14).

```
               XEQ "FULLHYD"

=======================
=======================
DEPTH<M>?
              120    RUN

SET WEIGHT TO SURFACE-
SET METER WHEEL TO 0-
ATTACH FIRST BOTTLE-


BOTTLE   METER
ATTACH.  WHEEL   TARGET
 ORDER  READING DEPTH
------- ------- ------

   1        0      114
   2       14      100
   3       39       75
   4       64       50
   5       84       30
   6       94       20
   7      104       10
          119

LOWER CAST TO FINAL
METER WHEEL VALUE OF
119 M.


=======================
=======================
```

Appendix 2; this can be used to create a working program for other calculators or computers. Appendix 3 presents the algorithm in a form suitable for "hand" determination; that is, in a form to allow precise meter wheel locations to be determined quickly and confidently without the aid of a computer.

Use at sea

Our calculator-based algorithm has been welcomed by both highly experienced and other research personnel as a valuable addition to the hydro lab. All recognize the need to ensure consistent performance at sea and the advantages of having the bottle placement, etc., determined automatically.

Other features such as limits to the number of bottles per cast, automatic station and set labelling, time (GMT and LOCAL) and date features, and even routines to announce when the messenger may be sent, could all be added. But these elaborations are not necessary for the benefits of machine-generated tables of meter wheel values and depths to substantially improve the conduct of full hydro casts. Appendices 1 and 2 should be helpful to others in this respect. But the explicit algorithm is also a powerful and easily understood aid when calculations must be done by hand; Appendix 3 provides point-by-point steps to derive, by hand, the meter wheel values in outline form as well as the form of an efficient worksheet (Fig. 14).

Based on our limited experience of this routine, the use of the algorithm in calculator/computer and/or hand-determination forms will streamline and otherwise improve the conduct of full hydro stations, and should be considered for formal incorporation into standard cruise protocols.

Acknowledgements

Jeanine Hurley typed the manuscript under trying conditions. F. Cunningham and W. G. McMullon provided photographic and drafting services. Several people have encouraged me to complete the algorithm in its present general form, and provided helpful comments on the manuscript, especially M.J. Power and R. I. Perry. Dr. G. Lacroix kindly translated the abstract. The enthusiasm of T. D. Iles for the benefits accruing from the use of the algorithm is very much appreciated as are his contributions to the Introduction.

References

Defant, A. 1961. Volume I of physical oceanography. Pergamon Press, New York. 729 p.

Koeller, P. 1981MS. Manual for groundfish survey personnel - cruise preparation, conduct and standing orders. Can. Dep. Fish. Oceans Mar. Fish. Div. Lab. Reference No. 81/3 :79 p.

Perry, R.I. 1987MS. Notes on hydrographic procedures for Needler cruises - July 1987. Manuscript :14 p.

Subsidiary Committee on Oceanography. 1932. Physics of the earth - V. Oceanography. Bull. Natl. Res. Counc. 85 :581 p.

Sverdrup, H.U., M.W. Johnson, and R.H. Fleming. (Editor.) 1942. The oceans; their physics, chemistry, and general biology. Prentice-Hall, Inc., Englewood Cliffs, N.J. 1087 p.

U.S. Hydrographic Office. 1955. Instruction manual for oceanographic observations. Second Edition. U.S. Hydrographic Office, Secretary of the Navy, Washington. 210 p.

Weyl, P.K. 1970. Oceanography, an introduction to the marine environment. John Wiley & Sons, Inc., New York. 535 p.

International Council for the Exploration of the Sea

C.M. 1987/C:14 Appendix 1
Hydrography Committee

AN ALGORITHM FOR DETERMINING THE APPROPRIATE METER WHEEL VALUES FOR SAMPLING OF STANDARD DEPTHS AT HYDROGRAPHIC STATIONS

APPENDIX 1

A WORKING PROGRAM FOR THE HP41C

by

J. B. Sochasky
Biological Sciences Branch
Marine Fish Division
Department of Fisheries and Oceans
Biological Station
St. Andrews, New Brunswick EOG 2X0
Canada

The program listings presented in Figures 5 to 8 provide a complete working program for the HP41C. The portions issuing printer commands (PRINTX2, HEAD, TAIL), the storage register terminology and the means of controlling branching and looping may be unique to the HP41C family and similar HP calculators; but the logical sequence should be easily translatable for any programmable calculator that uses RPN (reverse Polish notation), if subroutines are supported.

I will attempt to provide these programs on magnetic cards to anyone who sends me the necessary 25 magnetic cards (HP part #00097-13141) and a self-addressed, return envelope.

Principle Features

A glossary of program names is provided in Table 2.

The user normally interacts with only two programs, FHINIT and FULLHYD, which are assigned for single keystroke execution to seldom used keys on the HP41C keyboard (the "LN" and "TAN" keys on the right-hand side in Fig. 4).

The programs provide nice tabular output when the calculator is connected to a printer (Fig. 3).

When no printer is connected, however, both the initializing (FHINIT) and main (FULLHYD) programs automatically switch to a mode in which each line of output is frozen on the calculator display until the R/S ("run/stop") key is pushed.

Fig. 4. The HP41C algorithm produces an alternate line-by-line display on the calculator when no printer is attached. This line of display corresponds to bottle number 2; compare with the format produced by the printer (Fig. 3); see also Fig. 1 and the example on the "worksheet" (Fig. 14).



In addition, both programs begin by first attempting to print to the printer; if one is connected but turned off, this attempt results in a "PRINTER OFF" error and program interruption. As presently written, the two programs "crash" promptly with this error before the user has keyed in any data. Never connect or disconnect the calculator and printer unless both are turned off.

FHINIT must be run once to store the values for "distance between the hydro lab and surface," "roll compensation" and "depth limit"; the respective default values of 5, 1 and 400 m are stored when the R/S key is hit without first entering some other value.

FULLHYD prompts for the sounding depth. It makes extensive use of other subroutines to perform the data manipulations though. FULLHYD itself calculates the values for the deepest bottle. FULLHYD will accept as the "default" sounding depth, the value most recently used; i.e. the R/S key can be used without first keying in a repetition of the previous sounding depth.

The reminders and double lines printed at the top of the tabular output (Fig. 3) help delineate the data and ensure no initial steps are forgotten. They do take time to print however. I have included optional suppression of the printing of the double lines and reminders, but recommend suppression only for the most impatient of personnel:

- suppress double lines: Set Flag 01
- suppress reminders : Set Flag 00

The example given in the worksheet application (Appendix 3) and the situation shown in Fig. 1, correspond to the tabular example output shown in Figure 3. All of these consider a 120 m sounding.

A description of the algorithm

Before proceeding with explicit details, the algorithm can be summarized in this way. Three columns of data make up the table and represent 3 variables: BOTTN for bottle number, MWHEEL for meter wheel reading, and TDEPTH for target depth. The first two start at 1 and Ø, respectively. BOTTN increments by 1 for each line of the table. MWHEEL is incremented by the appropriate standard interval for each line of the table except for the interval between bottles 1 and 2 (see Fig. 1 and Table 1). The deepest TDEPTH depends on other variables: either the depth limit (like 400 m) or the sounding less clearances (e.g. 5 m and 1 m) in the cruise protocol. Other TDEPTHS are coded explicitly in the program.

The algorithm begins by initializing three "protocol" variables using "FHINIT" (Full Hydro INITialize): it prompts for and stores values for:

- the "distance between the hydro lab and surface in meters" (default 5 m),

Table 1. Standard target depths and meter wheel intervals for hydrographic observations (U.S. Hydrographic Office 1955).

Target depth (m)	Interval from next shallower depth (m)	Target depth (m)	Interval from next shallower depth (m)
10		800	100
20	10	1000	200
30	"	1200	"
50	20	1500	300
75	25	2000	500
100	"	2500	"
150	50	3000	"
200	"	4000	1000
250	"	5000	"
300	"	6000	"
400	100	7000	"
500	"	8000	"
600	"	9000	"
700	"	10000	"
		11000**	"

**The deepest ocean point is in the Marianna Trench at 10,863 m (Sverdrup et al. 1942).

- the "roll compensation in meters" (default 1 m) (see Fig. 1),

- the "depth limit" below which no samples are required for the given survey (default 400 m).

FHINIT must be run at least once to store the values for these variables appropriately. Thereafter, "FHINIT" need be run only to alter those values. We'll assume here that default values have been used. The actual table of bottle numbers, etc. is generated by running "FULLHYD" (FULL HYDro) which first prompts for the depth (i.e. the sounding) in meters. In our HP41C application both "FHINIT" and "FULLHYD" are assigned to seldom used keys on the keyboard and subsequently executed with single key strokes (Fig. 4).

The station's bottom depth is stored; from it are subtracted the values for distance from hydro lab to surface (5 m) and roll compensation (1 m) and this value (the Sounding Less Lab and Roll, say SLLR, in this example) is tested against the protocol depth limit (say 400 m).

If the limit is greater (deeper) than the SLLR, then the SLLR will become the target depth (TDEPTH) of the first or bottom bottle, the one which will be attached to the wire when the weight is just in the water and the meter wheel is zeroed; in this case FULLHYD prints reminders to this effect as well as the four heading lines of the table.

If the depth limit is less (shallower) than the SLLR, then the value for the limit replaces the SLLR value in storage before the reminders are printed and the table started; this ensures that the deepest bottle is sent to the depth limit, e.g. 400 m, and not deeper.

FULLHYD has by this time also stored the value of Ø for the current bottle number (BOTTN) and current meter wheel reading (MWHEEL).

At this point, FULLHYD calls the subroutine "PRINTX." This first increments BOTTN by 1, then subtracts the target depth for the current bottle from the target depth for the <u>deepest</u> bottle, giving the meter wheel reading (MWHEEL) (which when the "current" bottle is also the first, the deepest bottle, is of course zero)*. PRINTX then calls a further subroutine (PRINTX2) which actually prints the values for bottle number, meter wheel reading and target depth. PRINTX2 contains loops to adjust the spacing according to the number of digits in each value but does not update any of the values themselves.

The data for the first (deepest) bottle having been printed, control returns to FULLHYD where a choice must be made as to which of several subroutines to call next, which choice depends on the depth. This part of the algorithm is complicated by the division of the possible depths into groups based on depth ranges. The subroutine "XLT50" (X less than 50 m) handles target depths of 30, 20, and 10 meters; XLT150 (X less than 150 m) handles target depths of 100, 75, and 50 meters; XLT400 (XLT less than 400 m) handles target depths 300, 250, 200, and 150 meters, and so on.

---

*This rather cumbersome logic is necessary to determine the interval, and hence MWHEEL, associated with bottle 2. It provides the correct MWHEEL value for all bottles, although the progressive addition of intervals is the intuitively and manually (Appendix 3) simpler approach.

Since we expect more shallow stations than deep stations, the algorithm tests for the shallower depth ranges first to ensure the shortest possible mean computation time. That is, SLLR is compared first to 50; if 50 is greater than SLLR (i.e. a shallow station with the "sounding less the default 6 meters" less than 50) control passes directly to XLT50 for the depths 30, 20, 10 to be considered. If the station is deeper, the SLLR is next compared to 150 for the next most likely range of depths (100, 75, 50) and XLT150 is called if appropriate. If not, SLLR is compared to 400 and XLT400 (for the specific depths 300, 250, 200, 150) is called if the depth is in this range. If the depth is deeper, the process continues with testing for successively deeper ranges [XLT1000 (800, 700, 600, 500, 400), XLT4000 (2500, 2000, 1500, 1200, 1000) or XLT11M (from 3000-11000**)] until the range containing the second bottle is found.

Within each of the "XLTnnn" subroutines, the explicit target depths are individually compared to the SLLR value by another subroutine (to ensure that the given specific depth is not deeper than the sounding) and if appropriate, PRINTX is called to print the bottle number, meter wheel and target depth; "PRINTX" again updates the meter wheel value and bottle number. The next shallower specific depth of the given range is then compared and printed if appropriate, and so on. When the last depth of the range for a given XLTnnn subroutine printed, control passes to the XLTnnn subroutine corresponding to the next shallower range and so on until the last bottle representing the shallowest target depth of the shallowest range (i.e. the 10 m depth) is printed. Then a final reminder is printed to lower the entire cast several more meters (usually 15: i.e. 10 m deep plus the "default 5 m," we've assumed, between the hydro lab and the water surface). The final meter wheel reading is thus the depth of the deepest bottle plus 5 m (in the default situation), or expressed another way, it is the sounding depth less the value for roll compensation (usually 1 m).

How Long Does It Take

The algorithm asks the calculator to format and manipulate alphanumerical calculations, make several comparisons, as well as update several variables. The slowest part of the process is the delay between when a long string has been printed on paper and when the same string has finished rolling across the calculator's display.

For the example of Figure 3 (120 m sounding), the program requires: 15 sec to print the double lines and issue the prompt; 15 sec to print the reminders; 25 sec more to complete the header lines of the table; only 6 sec to calculate and print the data for the first bottle; 1 min 25 sec more to finish the three columns; and 20 sec. to complete the final reminder. This

**The maximum standard depth provided in the algorithm is 11000 m, deeper than the Mariana Trench. In our own cruises on the continental shelf and slopes of eastern Canada, it is exceptional for the depths encountered to be over 400 m and even more exceptional for the depth limit for sampling to be deeper than 400 mm. The subroutines for the deeper depths would never be called under these circumstances and could be cleared from memory.

is a total of 2.5 min from the prompt to the finished table for a fully recharged printer and calculator. With no printer attached, the time from prompt to bottle 1 data is about 20 sec; use the suppression options and this drops to 4 sec.

The experienced technician could compete quite favorably with some of these times, especially if he used an efficient worksheet (Fig. 14). However, the time saving potential for other uses, as well as the comparative certainty of the calculated results are important considerations.

Our groundfish surveys have as many as 100 stations, 20 of which have full hydro casts; if say 5 min could be saved per cast (not an experienced technician), perhaps 1.5 hours could be saved on the cruise, an amount of time approaching that required to steam from one station to the next. Considering the high cost of vessel time, the cost and effort to automate might be economic on a quantitative (more data) basis.

Hardware Requirements

The full algorithm presented here requires an HP41C with a "Quad" memory expansion module (or an HP41CV or HP41CX) and a companion printer. A card reader, rechargeable battery packs, and rechargers are practical necessities; in the case of a "Memory Lost" condition, the entire set of programs would have to be rekeyed by hand unless both a card reader and the programs stored on cards were available. The "Memory Lost" condition can occur by accident, as well as when the batteries in the battery pack are changed too slowly. The "black type" paper tape is usually preferable for long-term records, where these are desired, as the blue usually photocopies poorly.

Possible Problems on the HP41C

Spurious Results

The algorithm uses 15 storage registers (numbers 00 to 14). If you have other programs which update these registers or if you manually change their contents, you could get spurious resuts. Soluton: run FHINIT again.

No sounding Recorded on Paper Tape

The "DEPTH <M> ?" prompt and the value, if any, you key in will be printed under the top pair of double lines only if the printer is set to the NORM condition. The MAN position results in the loss of this part of the printed record.

The FHINIT and FULLHYD Function Keys Don't Work

The assignment of program names to seldom used keys for single keystroke operation is effective only when the calculator is in USER mode. Ensure the calculator is in USER mode.

Printer and Display Formats

The algorithm checks flags 21 and 55 to see whether a printer is attached as well as to control whether the output is suitable for printing to paper or simply for line-by-line display on the calculator. Turn the calculator off and back on, then repeat FULLHYD if you get the wrong format (see also below). Never connect or disconnect the printer and calculator unless both are off.

Clearing the Options

As noted, the algorithm allows optional suppression of the initial double lines and reminders. To get them back, execute the commands "CF 01" and "CF 00" to clear the respective flags.

Slow Response

Do not expect the printer to remain at peak performance for an entire cruise. Use rechargeable battery packs and recharge them regularly or, better still, arrange a mounting location where continuous recharging is possible.

Other Problems

If possible, go to sea with spares for all pieces of hardware (calculator, printer, card reader, charger, battery pack, etc.), magnetic cards (a complete "write all" set as well as cards with individual programs), paper rolls (preferably black), printed listings of each program, and manuals for each piece of hardware.

When all else fails, turn the hardware off and on again and retry FHINIT; replace programs you suspect of having been corrupted, or replace all of them; refer to the HP manuals. If corruption is a recurring problem, (someone unwittingly goes into "program mode" and accidentally alters a program) prepare the magnetic cards with the "private" security feature (see the card reader manual).

Fig. 5. The FHINIT and FULLHYD routines which initialize required storage registers, establish the target depth for the deepest bottle and control all the other subroutines.

```
01♦LBL "FHINIT"
02 "VERS 1.1"
03 XEQ "PRTROFF"
04 FC? 55
05 SF 21
06 5
07 "DISTANCE"
08 AVIEW
09 "BETWEEN"
10 AVIEW
11 "HYDRO LAB"
12 AVIEW
13 "AND SURFACE"
14 AVIEW
15 "IN METRES?"
16 PROMPT
17 STO 00
18 1
19 "ROLL"
20 AVIEW
21 "COMPENSATION"
22 AVIEW
23 "IN METRES?"
24 PROMPT
25 STO 01
26 400
27 "DEPTH LIMIT?"
28 PROMPT
29 STO 11
30 " "
31 ASTO 07
32 "  "
33 ASTO 08
34 "   "
35 ASTO 09
36 CLX
37 END
```

```
01♦LBL "FULLHYD"
02 "V PRNTR 3.2"
03 XEQ "PRTROFF"
04 FS? 01
05 GTO 05
06 XEQ "LINES"
07♦LBL 05
08 RCL 03
09 "DEPTH<M>?"
10 PROMPT
11 STO 03
12 RCL 00
13 -
14 RCL 01
15 -
16 STO 02
17 STO 06
18 0
19 STO 04
20 RCL 02
21 RCL 11
22 X>Y?
23 GTO 07
24 STO 02
25 STO 06
26♦LBL 07
27 ADV
28 XEQ "HEAD"
29 FIX 0
30 ADV
31 XEQ "PRINTX"
32 RCL 02
33 50
34 X<=Y?
35 GTO 08
36 GTO 01
37♦LBL 08
38 RCL 02
39 150
40 X<=Y?
41 GTO 09
42 GTO 02
43♦LBL 09
44 RCL 02
45 400
46 X<=Y?
47 GTO 10
48 GTO 03
49♦LBL 10
50 RCL 02
51 1000
52 X<=Y?
53 GTO 11
54 GTO 04
55♦LBL 11
56 RCL 02
57 3000
58 X<=Y?
59 XEQ "XLT11M"
60 XEQ "XLT3000"
61♦LBL 04
62 XEQ "XLT1000"
63♦LBL 03
64 XEQ "XLT400"
65♦LBL 02
66 XEQ "XLT150"
67♦LBL 01
68 XEQ "XLT50"
69 XEQ "TAIL"
70 XEQ "LINES"
71 END
```

Fig. 6. The PRTROFF, LINES, HEAD and TAIL subroutines which print or display the various reminder, column heaidngs, etc. in the table as well as determine the final MWHEEL value.*

```
01♦LBL "PRTROFF"
02 "VERS 1.0"
03 FC? 55
04 GTO 01
05 " "
06 PRA
07♦LBL 01
08 END
```

```
01♦LBL "LINES"
02 "VERS 1.0"
03 FC? 55
04 GTO 01
05 "============="
06 "├============"
07 AVIEW
08 "============"
09 "├============"
10 AVIEW
11 CLA
12♦LBL 01
13 END
```

```
01♦LBL "HEAD"
02 "VERS 1.0"
03 FS? 00
04 GTO 02
05 "SET WEIGHT TO"
06 "├ SURFACE-"
07 AVIEW
08 "SET METER WHEEL"
09 "├ TO 0-"
10 AVIEW
11 "ATTACH FIRST"
12 "├ BOTTLE-"
13 AVIEW
14♦LBL 02
15 FC? 55
16 GTO 01
17 ADV
18 ADV
19 "BOTTLE   METER"
20 AVIEW
21 "ATTACH.  WHEEL "
22 "├ TARGET"
23 AVIEW
24 " ORDER  READING"
25 "├ DEPTH"
26 AVIEW
27 "------- -------"
28 "├ ------"
29 AVIEW
30♦LBL 01
31 END
```

```
01♦LBL "TAIL"
02 "VERS 2.0"
03 "MWHEEL"
04 RCL 05
05 RCL 06
06 +
07 RCL 00
08 +
09 STO 05
10 STO 13
11 FC? 55
12 GTO 01
13 XEQ "SPACES"
14 ARCL 09
15 ARCL 09
16 ARCL 08
17 ARCL 14
18 AVIEW
19 ADV
20 "LOWER CAST TO"
21 "├ FINAL"
22 AVIEW
23 "METER WHEEL "
24 "├VALUE OF"
25 AVIEW
26 CLA
27 ARCL 05
28 "├ M."
29 AVIEW
30 ADV
31 ADV
32 ADV
33 GTO 02
34♦LBL 01
35 "FINAL MW "
36 "├"
37 ARCL 13
38 AVIEW
39♦LBL 02
40 END
```

*The "lazy T" symbol (i.e.├) refers to the APPEND function.

Fig. 7. The PRINTX, PRINTX2, SPACES and SPACESX subroutines which determine the format for printing or display and update values for BOTTN, and MWHEEL.

```
01♦LBL "PRINTX"
02 "VERS 1.1"
03 "BOTTN"
04 RCL 04
05 1
06 +
07 STO 04
08 "MWHEEL"
09 RCL 02
10 RCL 06
11 -
12 STO 05
13 XEQ "PRINTX2"
14 END
```

```
01♦LBL "PRINTX2"
02 "VERS 3.1"
03 FC? 55
04 GTO 12
05 "BOTTN"
06 CLA
07 RCL 04
08 10
09 X>Y?
10 GTO 01
11 GTO 02
12♦LBL 01
13 ARCL 09
14 GTO 03
15♦LBL 02
16 ARCL 08
17♦LBL 03
18 ARCL 04
19 ARCL 09
20 ASTO 12
21 "MWHEEL"
22 RCL 05
23 STO 13
24 XEQ "SPACES"
25 ARCL 14
26 ASTO 10
27 "TDEPTH"
28 RCL 06
29 STO 13
30 XEQ "SPACES"
31 ARCL 12
32 ARCL 08
33 ARCL 10
34 ARCL 09
35 ARCL 14
36 AVIEW
37 GTO 13
38♦LBL 12
39 XEQ "SPACESX"
40♦LBL 13
41 END
```

```
01♦LBL "SPACES"
02 "VERS 1.0"
03 CLA
04 RCL 13
05 10
06 X>Y?
07 ARCL 07
08 RCL 13
09 100
10 X>Y?
11 ARCL 07
12 RCL 13
13 1000
14 X>Y?
15 ARCL 07
16 RCL 13
17 10000
18 X>Y?
19 ARCL 07
20 ARCL 13
21 ASTO 14
22 CLA
23 END
```

```
01♦LBL "SPACESX"
02 "VERS 1.0"
03 SF 21
04 "B"
05 "├"
06 ARCL 04
07 "├ M"
08 "├"
09 ARCL 05
10 "├ D"
11 "├"
12 ARCL 06
13 AVIEW
14 END
```

Fig. 8. The XLT and XLTnnn family of subroutines which determine for which standard depths (TDEPTHs) the subroutine PRINTX is called.

```
♦LBL "XLT"
"VERS 2.0"
STO 06
RCL 02
X>Y?
XEQ "PRINTX"
END
```

```
♦LBL "XLT50"
"VERS 2.0"
30
XEQ "XLT"
20
XEQ "XLT"
10
XEQ "XLT"
END
```

```
♦LBL "XLT150"
"VERS 2.0"
100
XEQ "XLT"
75
XEQ "XLT"
50
XEQ "XLT"
END
```

```
♦LBL "XLT400"
"VERS 2.0"
300
XEQ "XLT"
250
XEQ "XLT"
200
XEQ "XLT"
150
XEQ "XLT"
END
```

```
♦LBL "XLT1000"
"VERS 2.0"
800
XEQ "XLT"
700
XEQ "XLT"
600
XEQ "XLT"
500
XEQ "XLT"
400
XEQ "XLT"
END
```

```
♦LBL "XLT3000"
"VERS 2.0"
2500
XEQ "XLT"
2000
XEQ "XLT"
1500
XEQ "XLT"
1200
XEQ "XLT"
1000
XEQ "XLT"
END
```

```
♦LBL "XLT11M"
"VERS 2.0"
11000
XEQ "XLT"
10000
XEQ "XLT"
9000
XEQ "XLT"
8000
XEQ "XLT"
7000
XEQ "XLT"
6000
XEQ "XLT"
5000
XEQ "XLT"
4000
XEQ "XLT"
3000
XEQ "XLT"
END
```

AN ALGORITHM FOR DETERMINING THE APPROPRIATE METER WHEEL VALUES FOR SAMPLING OF STANDARD DEPTHS AT HYDROGRAPHIC STATIONS

APPENDIX 2

A GENERAL FORM OF THE ALGORITHM

by

J. B. Sochasky
Biological Sciences Branch
Marine Fish Division
Department of Fisheries and Oceans
Biological Station
St. Andrews, New Brunswick E0G 2X0
Canada

The flowchart outlines (Fig. 9-Fig. 13) describe the algorithm, its subroutines and loops in a generalized fashion. It should serve as a starting point for the preparation of a working program for any suitable calculator or computer.

The division of the standard depths (Table 1) into depth ranges corresponding to the XLTnnn subroutines is arbitrary. On fast machines, the step of finding the depth range might even be omitted; in such a case the algorithm would be written to immediately locate the actual depth of BOTTN 2, 3, etc., after determining the TDEPTH for BOTTN 1.

See also the discussion in Appendix 1.

Figure 9. Flowchart outline for the FHINIT program.



Figure 10. Flowchart outline for the first portions of the FULLHYD program. The processing flow loops in and out of Fig. 11 and continues in Fig. 12.



Figure 11. Flowchart outline for the PRINTX and PRINTX2 subroutines. PRINTX is executed by FULLHYD for BOTTN 1 (Fig. 10) and by XLT (Fig. 12) for all other bottles.



* The HP41C program (Appendix 1) has two distinct formats, one for use with a printer is attached and one without (see Fig. 3, Fig. 4).

Figure 12. Flowchart outline for the final portions of the FULLHYD program. The XLTnnn subroutines are exemplified in more detail by XLT150 and XLT50 in Fig. 13. The XLTnnn subroutines loop in and out of XLT (Fig. 13).



Figure 13. Flowchart outline for the XLT and XLTnnn subroutines, as exemplified by XLT150 and XLT50. The XLTnnn subroutines call XLT for each standard depth [STAN]. In this case, the processing flow is shown to proceed from XLT150 to XLT50 as it would for any TDEPTH for BOTTN 2 deeper than 50 m.



International Council for the Exploration of the Sea

C.M. 1987/C:14 Appendix 3
Hydrography Committee

AN ALGORITHM FOR DETERMINING THE APPROPRIATE METER WHEEL VALUES FOR SAMPLING OF STANDARD DEPTHS AT HYDROGRAPHIC STATIONS

APPENDIX 3

THE HAND DETERMINATION OF METER WHEEL VALUES AT HYDROGRAPHIC STATIONS

by

J. B. Sochasky
Biological Sciences Branch
Marine Fish Division
Department of Fisheries and Oceans
Biological Station
St. Andrews, New Brunswick EOG 2XO
Canada

Determining the appropriate meter wheel values "by hand" is the traditional approach and will continue to be required when "machine methods" fail. Moreover, some highly experienced personnel may prefer hand determinations as they can be faster than a hand calculator, at least for completion of the entire table.

The outline below allows the straightforward step-by-step determination of the appropriate bottle positions by hand in a manner that parallels the calculator program provided in Appendix 1.

Preliminaries (Initialization)

1. Determine the distance from the surface to the hydro lab (usually about 5 m). If in doubt, zero the meter wheel with the weight in the position at which the bottles are normally attached and then lower it into the position it usually takes (i.e just below the surface) when the first bottle is normally attached; the value on the meter wheel is the required value.

2. Determine the value for roll compensation (normally 1 m) from the protocol for the cruise.

Procedures: Meter Wheel and Interval Determinations

Bottle 1 (the deepest bottle)

1. obtain the sounding;

2. set hydro weight just below surface;

3. zero meter wheel;

4. attached first (i.e. deepest) bottle;

5. determine target depth: if the sounding is deeper than the limit set by the cruise program, the target depth becomes the depth limit; otherwise, from the sounding, subtract the values for the distance from the surface to the hydro lab and that for roll compensation;

6. determine the interval between the first and second bottles: find the shallower standard depth (Table 1) closest to the target depth for bottle 1 (determined in step 5, above); the difference between them is the required interval;

7. lower the hydrographic wire and the first bottle until the meter wheel reads the required interval;

8. Proceed to Bottle 2.

Bottle 2

1. ensure that the meter wheel shows a value equal to the required interval between Bottles 1 and 2;

2. attach bottle 2;

3. proceed to the next shallower and subsequent bottles.

Subsequent Bottles

1. from the table of standard depths and intervals (Table 1), determine the target depth and interval to the next shallower bottle;

2. add this interval to the current meter wheel value;

3. lower the wire and bottles to this new value;

4. attach this bottle;

5. repeat this procedure for each subsequent bottle until the last bottle (i.e. the shallowest bottle, normally for the 10 m depth) has been attached.

The last (shallowest) bottle

1. Note that at this point the last bottle is on the wire with the meter wheel showing a value equivalent to the total interval between the top and bottom bottles.

2. The bottles must now all be lowered an additional amount until they reach their target depths.

3. Determine the final meter wheel reading required by adding to the total interval between the top and bottom bottles an amount equal to the sum of the target depth of the last (shallowest) bottle (normally 10 m) plus the distance between the surface and the hydro lab (normally 5 m).

4. Verify that this is the same as the sounding less the roll compensation (normally 1 m).

5. Lower the bottles to this final desired meter wheel value.

6. Note the time of day; the period of temperate equilibration at depth begins at this time.

A more abbreviated set of instructions (that don't consider a depth limit and consider only the default "protocol" values) is preprinted on the "worksheet" adapted from Perry (1987) and which greatly simplifies the procedure. The standard depths (to 500 m) and their intervals are included on the working copy and note that the form is to be used from the bottom up. The example printed there uses a sounding of 120 m which is the same as that for Figures 1 and 3. Note also that the only tedious determinations are those for the deepest bottle (the one first attached) and that once the target depth and interval for that bottle are recorded, the cumulative depths to each of the shallower bottles can be completed without further delay.

In our use, the worksheet is intended to be preprinted on the reverse side of the hydrographic station coding forms on which the temperatures are recorded, but if desired, one could incorporate such a table onto the main part of a coding form.

Table 2. A glossary of terms concerning the subroutines comprising the working algorithm for the HP41C family of programmable calculators.

BOTTN - This is a mnemonic term for BOTTle Number; it is not a subroutine.

FHINIT - Full Hydro INITialize; a routine which prompts and stores values for 3 variables in their appropriate registers.

FULLHYD - FULL HYDro; the main program which prompts for the sounding depth and calls other subroutines to calculate and display or print the bottle number, meter wheel readings and target depth, as well as table headings, reminders, etc.

HEAD - This routine controls the printing for the table's HEADing, that is the reminders and the column labels. (Reminders suppressed if flag 00 is set.)

LINES - The routine which normally prints the pair of double LINES at the top and bottom of the table. (The top pair are suppressed if flag 01 is set.)

MWHEEL - This is a mnemonic term for Meter WHEEL; it is not a subroutine.

PRINTX - PRINT table values in line X; this subroutine is called by XLT and, after updating BOTTN and MWHEEL, in turn calls PRINTX2.

PRINTX2 - PRINT table values in line X, subroutine 2; this subroutine is called by PRINTX. If there is no printer attached, it calls SPACESX and then ends; if there is a printer attached, it prepares the values of BOTTN, MWHEEL and TDEPTH for printing by right-justifying them, using the subroutine SPACES in the latter two cases.

PRTROFF - PRinTeR OFF; a routine called at the beginning of FHINIT and FULLHYD to cause a "PRINTER OFF" interupt immediately if a printer is connected but not turned on, rather than later after prompts have been answered. It prevents a simple oversight from consuming unnecessary time.

SLLR - This is a mnemonic term for Sounding Less Lab and Roll; it is not a subroutine. This is normally the TDEPTH for BOTTN 1; it is used to help calculate the MWHEEL value.

SPACES - Insert SPACES to right justify; this subroutine inserts blank characters in front of given values to provide a tidy table format on the printer; it is called by PRINTX2.

SPACESX - An alternate to SPACES in line X; the routine is called by PRINTX2 when no printer is attached. It prefixes BOTTN, MWHEEL and TDEPTH with B, M and D, respectively, for display on a line by line basis (Fig. 4).

TAIL - This subroutine controls the "TAILer" or last lines of the table. It's called by FULLHYD, calcualtes the final meter wheel value, and arranges it on a format for printing or simply display, depending on the presence of a printer.

TDEPTH - This is a mnemonic term for Target DEPTH; it is not a subroutine.

XLT - This subroutine is called by all the XLTnnn (X Less Than) subroutines. It compares the current standard depth (located in the XLTnnn subroutines) with the TDEPTH for the bottom bottle. If the current depth is inappropriate, the sub-routine simply ends; if the current depth is to be printed, PRINTX is called.

XLT50 - This is the X Less Than 50 m subroutine for the standard dpeths of 30, 20 and 10 m. It is called by FULLHYD and in turn calls XLT.

XLT150 - This the X Less Than 150 m subroutine for the standard depths of 100, 75 and 50 m. It is called by FULLHYD and in turn calls XLT.

XLT400 - This is the X Less Than 400 m subroutine for the standard depths of 300, 250, 200 and 150 m. It is called by FULLHYD and in turn calls XLT.

XLT1000 - This is the X Less Than 1000 m subroutine for the standard depths of 800, 700, 600, 500 and 400 m. It is called by FULLHYD and in turn calls XLT.

XLT3000 - This is the X Less Than 3000 m subroutine for the standard depths of 2500, 2000, 1500, 1200 and 1000 m. It is called by FULLHYD and in turn calls XLT.

XLT11M - This is the X Less Than 11 thousand (Roman numeral M) M subroutine for the standard depths of 11000 m to 3000 m all with 1000 m intervals. It is called by FULLHYD and in turn calls XLT.

Figure 14. Worksheet for Full Hydrocast

Cruise #	Set #	Date	Time

1. Note: Use bottom portions of the table first as indicated.
2. Zero meter wheel with weight just be low surface; attach first bottle.
3. Determine target depth for bottom bottle (first attached)
   (a) su btract 6 m from sounding,
   (b) enter this value under "Target Depth" in the table just above the next shallower standard depth*,
   (c) record the interval to the next shallower standard depth under "Interval,"
   (d) record 0 under corresponding "cumulative meter wheel reading,"
   (e) record 1 under "Bottle #"
4. Determine the remaining meter wheel values by summing the intervals for the shallower standard depths and record these and their bottle numbers.

Bottle #	Target Depth (m)	Interval	Cumulative Meter Wheel Reading
	500	100	
	400	100	
	300	50	
	250	50	
	200	50	
	150	50	
	100	25	
	75	25	
	50	20	
	30	10	
	20	10	
	10	15	

Distance from hydro lab to shallowest bottle

5. The last entry in the interval column should be the distance between the shallowest bottle and the surface plus the distance from the surface to hydro lab.
6. Check the wire angle and correct if necessary.

*The standard depths 10 to 500 m and their corresponding intervals are indicated in small type in the table; the deeper standard depths are 600, 700 800, 1000, 1200, 1500, 2000 m and thereafter every 1000 m to the bottom.

Example:

For a sounding of 120 m:

- the target depth for the bottom bottle is 120-6 = 114 m.
- its corresponding interval to the next bottle is 114-100 = 14 m.

Bottle #	Target Depth (m)	Interval	Cumulative Meter Wheel Reading
		50	
	200	50	
1	150 114	50 14	0
2	100 100	25 25	14
3	75 75	25 25	39
4	50 50	20 20	64
5	30 30	10 10	84
6	20 20	10 10	94
7	10 10	15 15	104
			119

Distance from hydro lab to shallowest bottle

Final meter reading (uncorrected for wire angle)

(from Perry, 1987)